
Overview

About This Manual

Overview

Thismanual isintended as afirst introduction to RAPID. A lot of functionality in RAPID is
left out, but the most essential parts are described so that it can be easily understood for

everybody. Thismanual does not make you an expert RAPID programmer, but it can help you
understand the concept of programming with RAPID. The details can aways be found in the

reference manuals.

Usage

This manual should be read before starting to program. It does not contain everything you
need to know, but you need to be familiar with most things in this manual, before starting to

write aRAPID program.

This manual does not replace the educational coursesin RAPID, but can complement it.

Who Should Read This Manual?

Thismanual isintended for someone with no previous experience in programming, e.g. a
robot operator who wantsto learn how to program the robot.

Prerequisites

There are no prerequisites for this manual.

Organization of Chapters

The manual is organized in the following chapters:

Chapter
1. RAPID basics

2. RAPID robot functionality

3. Structure

Data with multiple values

RAPID instructions and
functions

6. What to read next

Contents

The fundamentals of programming. This functionality is
similar in most high level programming languages.

Describes the functionality that makes RAPID unique,
i.e. move instructions, I/0O signals and communication
with a FlexPendant.

Describes how to create procedures. Also contains a
brief introduction to how to apply a structured design of a
program.

Describes arrays and complex data types.

A short explanation of what the RAPID instructions and
functions are.

Where to find more information if you want to continue
your studies of RAPID.

References

Reference

Technical reference manual - RAPID overview
Technical reference manual - RAPID Instructions, Functions and

Data types

Technical reference manual - RAPID kernel
Operating manual - IRC5 with FlexPendant

Document Id

3HAC16580-1
3HAC16581-1

3HAC16585-1
3HAC16590-1
Continues on next page

Product documentation, M2004

Product documentation, M2004

General
The robot documentation is divided into a number of categories. Thislisting is based on the
type of information contained within the documents, regardless of whether the products are
standard or optional. Thismeansthat any given delivery of robot productswill not contain all
documents listed, only the ones pertaining to the equipment delivered.

However, all documents listed may be ordered from ABB. The documentslisted are valid for
M2004 robot systems.

Product manuals
All hardware, robots and controllers, will be delivered with a Product manual that contains:
» Sofety information

» Installation and commissioning (descriptions of mechanical installation, electrical
connections)

» Maintenance (descriptions of all required preventive maintenance procedures
including intervals)

» Repair (descriptions of all recommended repair procedures including spare parts)
» Additional procedures, if any (calibration, decommissioning)

» Reference information (article numbers for documentation referred to in Product
manual, procedures, lists of tools, safety standards)

o Partlist
» Foldouts or exploded views
e Circuit diagrams

Technical reference manuals

The following manuals describe the robot software in general and contain relevant reference
information:

» RAPID Overview: An overview of the RAPID programming language.

* RAPID Instructions, Functions and Data types: Description and syntax for all
RAPID instructions, functions and data types.

e System parameters: Description of system parameters and configuration workflows.

Application manuals
Specific applications (for example software or hardware options) are described in
Application manuals. An application manual can describe one or several applications.

An application manua generally contains information abouit:
» The purpose of the application (what it does and when it is useful)

»  What isincluded (for example cables, 1/0 boards, RAPID instructions, system
parameters, CD with PC software)

» How to use the application

» Examples of how to use the application

Continues on next page

Product documentation, M2004

Continued

Operating manuals

Thisgroup of manualsisaimed at those having first hand operational contact with the robot,
that is production cell operators, programmers and trouble shooters. The group of manuals
includes:

Emergency safety information

Getting started - IRC5 and RobotStudio
IRC5 with FlexPendant

RobotSudio

Trouble shooting - IRCS5 for the controller and robot

3HAC029364-001 Revision: -

Safety

Safety

Safety of personnel
A robot is heavy and extremely powerful regardless of its speed. A pause or long stop in
movement can be followed by afast hazardous movement. Even if a pattern of movement is
predicted, a change in operation can be triggered by an external signal resulting in an
unexpected movement.

Therefore, it isimportant that all safety regulations are followed when entering safeguarded
space.

Safety regulations

Before beginning work with the robot, make sure you are familiar with the safety regulations
described in Operating manual - IRC5 with FlexPendant.

3HAC029364-001 Revision: - 9



Terminology

Terminology

About the terms
Thismanual is generally written for beginners, regarding both programming and robots.
However, someterms are used that may befamiliar only to those with some knowledge about
programming and/or industrial robots. These terms are described in this terminol ogy.

Terms

Term Description

FlexPendant A hand held terminal for controlling a robot system.
Robot controller  The robot controller is basically a computer that controls the robot.

Syntax Rules for how a language is allowed to be written. It can be seen as the
grammar of the programming language.

The syntax of a programming language is much more strict than in
ordinary human language. Humans are intelligent and would understand
if | say "l fast run" instead of "I run fast". Computers, on the other hand,
are stupid and would not understand anything unless the syntax is
absolutely correct.

10 3HAC029364-001 Revision: -

1 RAPID basics
1.1. About RAPID

1 RAPID basics

1.1. About RAPID

What is RAPID

If you want a computer to do something, a program is required. RAPID is a programming
language for writing such a program.

The nativelanguage of computersconsists of only zerosand ones. Thisisvirtually impossible
for humans to understand. Therefore computers are taught to understand a language that is
relatively easy to understand - ahigh level programming language. RAPID isahigh level
programming language, it uses some English words (like IF and FOR) to make it
understandable for humans.

Simple RAPID program example
Let uslook at asimple example to see what a RAPID program can look like:

MODULE MainModule
VAR num length;
VAR num width;

VAR num area;

PROC main ()
length := 10;
width := 5;
area := length * width;
TPWrite "The area of the rectangle is " \Num:=area;

END PROC

ENDMODULE
This program will calculate the area of arectangle and write on the FlexPendant:

The area of the rectangle is 50

3HAC029364-001 Revision: - 11



1 RAPID basics

1.2.1. Variables

1.2 RAPID data

1.2.1. Variables

Data types

There are many different data typesin RAPID. For now, we will focus on the three general
data types:

Data type Description

num Numerical data, can be both integer and decimal number. E.g. 10 or
3.14159.

string A text string. E.g. "This is a string". Maximum of 80 characters.

bool A boolean (logical) variable. Can only have the values TRUE or FALSE.

All other datatypes are based on these three. If you understand them, how to perform
operations on them and how they can be combined to more complex datatypes, you can easily
understand all data types.

Variable characteristics

A variable contains a data value. If the program is stopped and started the variable keepsiits
value, but if the program pointer is moved to main the variable data value is | ost.

Declaring a variable

Example

Declaring avariable is the way of defining a variable name and which data type it should
have. A variable is declared using the keyword var, according to the syntax:

VAR datatype identifier;

VAR num length;
VAR string name;

VAR bool finished;

Assigning values

A vaueis assigned to avariable using the instruction : =

length := 10;
name := "John"
finished := TRUE;

Notethat : = isnot an equal sign. It means that the expression to the right is passed to the
variable on the left. There can only be avariable to the left of : =
For example, the following is a correct RAPID code resulting in reg1 having the value 3:
regl := 2;
regl := regl + 1;
The assignment can be made at the same time as the variable declaration:

VAR num length := 10;
VAR string name := "John";
VAR bool finished := TRUE;

12

1 RAPID basics

1.2.2. Persistent variables

1.2.2. Persistent variables

What is a persistent variable
A persistent variable is basically the same as an ordinary variable, but with one important
difference. A persistent variableremembersthelast valueit wasassigned, even if the program
is stopped and started from the beginning again.

Declaring a persistent variable
A persistent variable is declared using the keyword pERS. At declaration an initial value

mustbe assigned.
PERS num nbr := 1;
PERS string stringl := "Hello";

Example
Consider the following code example:

PERS num nbr := 1;
PROC main ()
nbr := 2;
ENDPROC
If this program is executed, the initial value is changed to 2. The next time the program is
executed the program code will look like this:
PERS num nbr := 2;
PROC main ()
nbr := 2;
ENDPROC

3HAC029364-001 Revision: - 13



1 RAPID basics

1.2.3. Constants

1.2.3. Constants

What is a constant?
A constant containsavalue, just likeavariable, but thevalueisawaysassigned at declaration
and after that the value can never be changed. The constant can be used in the program in the
same way as avariable, except that it is not allowed to assign a new value to it.

Constant declaration
The constant is declared using the keyword consT followed by data type, identifier and
assignment of avalue.
CONST num gravity := 9.81;
CONST string greating := "Hello"

Why use constants?
By using a constant instead of a variable, you can be sure that the value is not changed
somewhere in the program.

Using a constant instead of writing the value directly in the program is better if you need to
update the program with another value on the constant. Then you only haveto changein one
place and can be sure you have not forgotten any occurrence of the value.

14 3HAC029364-001 Revision: -

© Copyright 2007 ABB. All rights reserved.



© Copyright 2007 ABB. All rights reserved.

1 RAPID basics

1.2.4. Operators

Numerical operators

1.2.4. Operators

These operators operate on the datatype num and return the datatypenum. |.e. in the examples
below, variables reg1, reg2 and reg3 are of datatype num.

Operator

+

Description

Addition

Subtraction
Unary minus

Multiplication
Division

Example

regl := reg2 + reg3;
regl := reg2 - reg3;
regl := -reg2;

regl := reg2 * reg3;
regl := reg2 / reg3;

Relational operators

These operators return the data type bool.

In the examples, reg1 and reg2 are data type num while f1ag1l iSbool.

Operator

<>

Description

equal to

less than

greater than

less than or equal to

greater than or equal to

not equal to

Example

flagl := regl = reg2;

flagl is TRUE if regl equals reg2

flagl := regl < reg2;

flagl is TRUE if regl is less than reg2
flagl := regl > reg2;

flagl is TRUE if regl is greater than reg2
flagl := regl <= reg2;

flagl is TRUE if regl is less than or equal to
regz

flagl := regl >= reg2;

flagl is TRUE if regl is greater than or equal to
reg2

flagl := regl <> reg2;

flagl is TRUE if regl is not equal to reg2

Logical operators are often used together with the 1F instruction. For code examples, see
Examples with logical conditions and IF statements on page 18.

String operator

Operator

+

Description

String concatenation

Example

VAR string firstname := "John";
VAR string lastname := "Smith";
VAR string fullname;

fullname := firstname + " " +
lastname;

The variable fullname will contain the string
"John Smith".

3HAC029364-001 Revision: -

1 RAPID basics

1.3.1. IF THEN

1.3 Controlling the program flow

1.3.1. IF THEN

About the program flow

The program exampleswe have seen so far are executed sequentially, from top to bottom. For
more complex programs, we may want to control which codeis executed, in which order, and
how many times. First we will have alook at how to set up conditions for if a program
seguence should be executed or not.

IF
The 1F instruction can be used when aset of statements only should be executed if a specified
condition is met.
If thelogical condition in the TF statement is true, then the program code between the
keywords THEN and ENDIF isexecuted. If the conditionisfalse, that codeis not executed and
the execution continues after ENDIF.
Example
In thisexample the string st ring1 iswritten on the FlexPendant if it is not an empty string.
If string1l isan empty string, i.e. contains no characters, then no action is taken.
VAR string stringl := "Hello";
IF stringl <> "" THEN
TPWrite stringl;
ENDIF
ELSE
An 1F statement can also contain program code to be executed if the condition isfalse.
If thelogical condition in the TF statement is true, then the program code between the
keywords THEN and ELSE is executed. If the condition is false, then the code between the
keywords ELSE and ENDIF is executed.
Example
In thisexamplethe string st ring1 iswritten on the FlexPendant if it is not an empty string.
If string1l isan empty string, then the text "The string is empty” is written.
VAR string stringl := "Hello";
IF stringl <> "" THEN
TPWrite stringl;
ELSE
TPWrite "The string is empty";
ENDIF
Continues on next page
16 3HACO029364-001 Revision: -

1 RAPID basics

1.3.1. IF THEN
Continued
ELSEIF
Sometimesyou have morethan two alternative program sequences. You can then USe ELSEIF
to set up severa alternatives.
Example

In this example different texts are written depending on the value on the variable t ime.

VAR num time := 38.7;

IF time < 40 THEN
TPWrite "Part produced at fast rate";
ELSEIF time < 60 THEN
TPWrite "Part produced at average rate";
ELSE
TPWrite "Part produced at slow rate";
ENDIF
Notethat since thefirst condition istruethe first text will be written. The two other textswill
not be written (even though it istrue that time isless than 60).

3HAC029364-001 Revision: - 17



1 RAPID basics

1.3.2. Examples with logical conditions and IF statements

1.3.2. Examples with logical conditions and IF statements

Example
Use the IF statement to determine which text to write on the FlexPendant. Write on the
FlexPendant which part is fastest to produce.
VAR string partl := "Shaft";
VAR num timel;
VAR string part2 := "Pipe";
VAR num time2;
PROC main/()
timel := 41.8;
time2 := 38.7;
IF timel < time2 THEN
TPWrite partl + " is fastest to produce';
ELSEIF timel > time2 THEN
TPWrite part2 + " is fastest to produce";
ELSE
TPWrite partl + " and " + part2 + " are equally fast to
produce";
ENDIF
ENDPROC
Example

If it takes more than 60 seconds to produce a part, write a message on the FlexPendant. If the
boolean variable full speed is FALSE the message will tell the operator to increase the
robot speed. If full speed iSTRUE, the message will ask the operator to examinethereason
for the dow production.

VAR num time := 62.3;
VAR bool full speed := TRUE;
PROC main ()

IF time > 60 THEN
IF full speed THEN
TPWrite "Examine why the production is slow";
ELSE
TPWrite "Increase robot speed for faster production';
ENDIF
ENDIF
ENDPROC

18

1 RAPID basics
1.3.3. FOR loop

1.3.3. FOR loop

Repeating a code sequence

Another way of controlling the program flow isto repeat a program code sequence a number
of times.

How does the FOR loop work
The following code will repeat writing "Hello" 5 times:
FOR i FROM 1 TO 5 DO
TPWrite "Hello";
ENDFOR
The syntax of the For statement is:

FOR counter FROM startvalue TO endvalue DO
program code to be repeated

ENDFOR
The counter does not have to be declared, but acts as a numeric variable inside the For loop.
Thefirst time the code is executed, the counter hasthe value specified by the startvalue. The
value of the counter isthen increased by 1 for each time the code is executed. The last time
the code executesiswhen the counter isequal to endvalue. After that, the execution continues
with the programming code after ENDFOR.

Using the counter value
The value of the counter can be used in the For loop.
For example, calculating the sum of al numbers from 1 to 50 (1+2+3+...+49+50) can be
programmed like this:

VAR num sum := 0;

FOR i FROM 1 TO 50 DO
sum := sum + 1i;
ENDFOR
It isnot alowed to assign avalue to the counter in the FOr loop.

3HAC029364-001 Revision: - 19



1 RAPID basics

1.3.4. WHILE loop

1.3.4. WHILE loop

Repeating with condition
The repeating of acode sequence can be combined with the conditional execution of the code
sequence. With the wHILE loop the program will continue repeating the code sequence as
long as the condition is true.

WHILE syntax

The syntax for thewHILE loopis:

WHILE condition DO

program code to be repeated

ENDWHILE
If the condition is false to begin with, the code sequence will never be executed. If the
condition is true, the code sequence will be executed repeatedly until the condition is no
longer true.

Example
The following program code will add numbers (1+2+3+...) until the sum reaches 100.

VAR num sum := 0;

VAR num i := 0;

WHILE sum <= 100 DO

i :=1 + 1;
sum := sum + 1;
ENDWHILE

Do not create eternal or heavy loops without wait instruction
If the condition never becomes false the loop will continue constantly and consume vast
amount of computer power. It is allowed to write an eternal loop, but then it must contain
some waiting instruction that allows the computer to perform other tasks in the meantime.

Heavy loops (with lots of calculations and writing on the FlexPendant, without move
instructions) can require some waiting instruction even if the number of loops are limited.

WHILE TRUE DO

! Some code

! Wait instruction that waits for 1 second
WaitTime 1;
ENDWHILE
Note that move instructions work as waiting instructions, since the execution does not
continue until the robot has reached its target.

20 3HAC029364-001 Revision: -

© Copyright 2007 ABB. All rights reserved.



© Copyright 2007 ABB. All rights reserved.

1 RAPID basics
1.4.1. General RAPID syntax rules

1.4 Rules and recommendations for RAPID syntax

1.4.1. General RAPID syntax rules

Semicolon
The general ruleisthat each statement ends with a semicolon.
Examples
Variable declaration:
VAR num length;
Assigning values:
area := length * width;
Most instruction calls:
MoveL pl0,v1000,fine,tool0;
Exceptions
Some special instructions do not end with a semicolon. Instead there are special keywordsto
indicate where they end.
Example of instructions that do not end with semicolon:
Instruction keyword Terminating keyword
IF ENDIF
FOR ENDFOR
WHILE ENDWHILE
PROC ENDPROC
These keywords are very important to create agood structure of aRAPID program. They are
thoroughly described later in this manual.
Comments
A line starting with I will not be interpreted by the robot controller. Use thisto write
comments about the code.
Example

! Calculate the area of the rectangle

area := length * width;

3HAC029364-001 Revision: - 21



1 RAPID basics

1.4.2. Recommendations for RAPID code

1.4.2. Recommendations for RAPID code

Capitalized keywords

RAPID is not case sensitive, but it isrecommended that al reserved words (e.g. VAR, PROC)
are written in capital letters. For a complete list of reserved words, see Technical reference
manual - RAPID overview.

Indentations

Example

To make the programming code easy to grasp, use indentation. Everything inside a proc
(between proc and ENDPROC) should beindented. Everything inside an 1F-, FOR- Ofr WHILE-
statement should be further indented.

When programming with the FlexPendant, the indentation is done automatically.

VAR bool repeat;

VAR num times;

PROC main ()
repeat := TRUE;
times := 3;
IF repeat THEN
FOR i FROM 1 TO times DO
TPWrite "Hello!";
ENDFOR
ENDIF
END PROC
Notethat it is easy to see where the 1F statement starts and ends. If you would have several
IF statements and no indentations, it would be virtually impossible to find which ENDIF
corresponds to which 1F.

22

2 RAPID robot functionality

2.1.1. Movel instruction

2 RAPID robot functionality

2.1 Move instructions

2.1.1. MovelL instruction

Overview
The advantage with RAPID isthat, except for having most functionality found in other high
level programming languages, it is specially designed to control robots. Most importantly,
there are instructions for making the robot move.

MovelL

A simple move instruction can look like this:

MoveL pl0, v1000, fine, tool0;
where:

* MoveL isaninstruction that movesthe robot linearly (in astraight line) from its current
position to the specified position.

» p1o specifiesthe position that the robot shall move to.

* v1000 specifiesthat the speed of the robot shall be 2000 mmy/s.

* fine specifiesthat the robot shall go exactly to the specified position and not cut any
corners on itsway to the next position.

* toolo specifiesthat it isthe mounting flange at the tip of the robot that should move
to the specified position.

MovelL syntax
Movel. ToPoint Speed Zone Tool;

ToPoint
The destination point defined by a constant of data type robtarget. When programming
with the FlexPendant you can assign a robtarget vaue by pointing out a position with the
robot. When programming offline, it can be complicated to calculate the coordinates for a
position.
robtarget Will be explained further later, in section Composite data types on page 44. For
now, let us just accept that the position x=600, y=-100, z=800 can be declared and assigned
like this:

CONST robtarget pl0 := [ [600, -100, 800], [1, 0o, 0, 0], [0, O, O,
0], [ 9E9, 9E9, 9E9, 9E9, 9E9, 9E9] ];

Speed
The speed of the movement defined by a constant of datatype speeddata. There are plenty
of predefined values, such as:

Predefined speeddata Value

vb 5 mm/s
v100 100 mm/s
v1000 1000 mm/s

Continues on next page

2 RAPID robot functionality

2.1.1. Movel instruction

Continued
Predefined speeddata Value
vmax Maximum speed for the robot
A complete list of predefined speeddata valuesisfound in Technical reference manual -
RAPID Instructions, Functions and Data types, section Data types and speeddata.
When using a predefined value, it should not be declared or assigned.
Zone
Specifies a corner zone defined by a constant of datatype zonedata. There are many
predefined values, such as.
Predefined zonedata Value
fine The robot will go to exactly the specified position
z10 The robot path can cut corners when it is less than
10 mm from ToPoint.
z50 The robot path can cut corners when it is less than
50 mm from ToPoint.
A complete list of predefined zonedata valuesisfound in Technical reference manual -
RAPID Instructions, Functions and Data types, section Data types and zonedata.
When using a predefined value, it should not be declared or assigned.
The following RAPID instructions will result in the robot path shown below:
MoveL pl0, v1000, z50, tool0;
MovelL p20, v1000, fine, tool0;
xx0700000358
Tool

Specifies the tool that the robot is using, defined by a persistent variable of datatype
tooldata. If awelding gun, glue gun or a pen is attached to the robot, we want to program
the ToPoint for the tip of thistool. Thisis done automatically if atooldata is declared,
assigned and used in the MoveL instruction.

toolo isapredefined tool, representing the robot without any tool mounted onit, and should
not be declared or assigned. Any other tool should be declared and assigned before being
used.

24 3HAC029364-001 Revision: -

© Copyright 2007 ABB. All rights reserved.



© Copyright 2007 ABB. All rights reserved.

2 RAPID robot functionality

2.1.2. Coordinate systems

2.1.2. Coordinate systems

Base coordinate system
The position that a move instruction moves to is specified as coordinates in a coordinate
system. If no coordinate system is specified, the position is given relative to the robot base
coordinate system (also called base frame). The base coordinate system hasitsorigin in the
robot base.

xx0700000397

Customized coordinate systems
Another coordinate system can be defined and used by move instructions. Which coordinate
system the move instruction shall use is specified with the optional argument \wob3.

MoveL pl0, v1000, z50, tool0 \WObj:=wobjl;

For information about how to define a coordinate system, see Technical reference manual -
RAPID Instructions, Functions and Data types, section Data types and wobjdata.
For more information about coordinate systems, see Technical reference manual - RAPID
overview, section Coordinate systems.

2 RAPID robot functionality

2.1.3. Examples with MovelL

2.1.3. Examples with MoveL

Draw a square

A robot isholding a pen above a piece of paper on atable. We want the robot to move the tip
of the pen down to the paper and then draw a square.

p10

@ P20

p40 ® p30
xx0700000362
PERS tooldata tPen := [ TRUE, [[200, 30], f[1, o, 0 ,01], [0.8,
[62, 0, 171, [1, 0, O, O], O, 011;
CONST robtarget pl0 := [ [600, -100, 800], [0.707170, O, 0.707170,
ol, [o, o, 0, 0], [ 9E9, 9E9, 9E9, 9E9, 9E9, 9E9] 1;
CONST robtarget p20 := [ [600, 100, 800], [0.707170, O, 0.707170,
ol, [0, 0, 0, 0], [ 9E9, 9E9, 9E9, 9E9, 9E9, 9E9] ];
CONST robtarget p30 := [ [800, 100, 800], [0.707170, 0, 0.707170,
o], [0, O, 0, 0], [ 9E9, 9E9, 9E9, 9E9, 9E9, 9E9] ];
CONST robtarget p40 := [ [800, -100, 800], [0.707170, O, 0.707170,
o], [o, 0o, 0, 0], [ 9E9, 9E9, 9E9, 9E9, 9E9, 9E9] 1;
PROC main ()
MoveL pl0, v200, fine, tPen;
MoveL p20, v200, fine, tPen;
MovelL p30, v200, fine, tPen;
MoveL p40, v200, fine, tPen;
MovelL pl0, v200, fine, tPen;
ENDPROC

Continues on next page

2 RAPID robot functionality

2.1.3. Examples with MoveL

Continued

Draw with corner zones

Draw the samefigure asin the previous example, but with acorner zone of 20 mm at p20 and
acorner zones of 50 mm at p4o0.

p10 @ p20

p40 @ @ p30

xx0700000363
VAR tooldata tPen :=

VAR robtarget p40 :=

PROC main ()
MoveL pl0, v200, fine, tPen;
MoveL p20, v200, z20, tPen;
MoveL p30, v200, fine, tPen;
MoveL p40, v200, z50, tPen;
MoveL pl0, v200, fine, tPen;
ENDPROC

2 RAPID robot functionality

2.1.4. Other move instructions

2.1.4. Other move instructions

Several move instructions

There are anumber of move instructionsin RAPID. The most common are Movel, MoveJ
and MovecC.

MovelJ

Example

MovedJ isused to move the robot quickly from one point to another when that movement does
not haveto bein astraight line.

UseMoved to movetherobot to apoint intheair closeto wheretherobot will work. A MoveL
instruction does not work if, for example, the robot base is between the current position and
the programmed position, or if the tool reorientation istoo large. Moved can always be used
in these cases.

The syntax of Moved is analog with MoveL.

Moved pl0, v1000, fine, tPen;

MoveC

Example

MoveC is used to move the robot circularly in an arc.

MoveL pl0, v500, fine, tPen;
MoveC p20, p30, v500, fine, tPen;
MovelL p40, v500, fine, tPen;

p20

p30 p40

p10
xx0700000364

2 RAPID robot functionality

2.1.5. Execution behavior in corner zones

2.1.5. Execution behavior in corner zones

Why the special execution in corner zones?

The execution of aprogram isusually carried out in the order the statements are written.

In the following example the robot first movesto p10, then calculatesthe value of reg1 and
then movesto p2o:

MoveL pl0, v1000, fine, tool0;

regl := reg2 + reg3;

MoveL p20, v1000, fine, toolO;
But now look at this example:

MoveL pl0, v1000, z50, tooloO;

regl := reg2 + reg3;

MovelL p20, v1000, fine, toolO0;
If the calculation of reg1 would not start until the robot was at p10, then the robot would
have to stop there and wait for the next move instruction. What actually happensis that the
code is executed ahead of the robot movement. reg1 is calculated and the robot path in the
corner zone is calculated before the robot reaches p1o.

How does this affect my program

Solution

In many cases the exact time of execution does not affect the program. There are however
examples of when it does affect the program.

If you want to draw alinewith aspray gun between p10 and p20, and write the program like
this:

MoveL pl0, v300, z10, tspray;

| Start spraying

SetDO dol, 1;

MoveL. p20, v300, zl1l0, tspray;

! Stop spraying

SetDO dol, O0;

MoveL p30, v300, fine, tspray;
The result may look something like this:

p10 e P20

P30

xx0700000387
If youwant to set signalsin corner zones, and not use £ine, then there are special instructions

for solving this, e.g. MoveLDO, TriggL and DispL. For more information about these
instructions, see Technical reference manual - RAPID Instructions, Functionsand Data types.

Continues on next page

2 RAPID robot functionality

2.1.5. Execution behavior in corner zones

Continued

Avoid wait instructions or heavy calculations after corner zone
The robot controller can calculate the robot movement in corner paths even if there are
instructions in between the move instructions. However, if there isawait instruction after a
move instruction with a corner zone, the robot will not be ableto handlethis. Use fine inthe
move instruction before await instruction.

Thereisalso alimitation asto how many (and complex) cal culations the robot controller can
calculate in between move instructions with corner zones. Thisis mainly a problem when
calling procedures after a move instruction with a corner zone.

30 3HAC029364-001 Revision: -

2 RAPID robot functionality

2.2 1/0 signals

2.2.1.1/0 signals

2.2.1. 1/0O signals

About signals

Signals are used for communication with external equipment that the robot cooperates with.
Input signals are set by the external equipment and can be used in the RAPID program to
initiate when to perform something with the robot. Output signals are set by the RAPID
program and signals to the external equipment that they should do something.

Setting up signals

Signals are configured in the system parameters for the robot system. It is possible to set
customized names for the signals. They should not be declared in the RAPID program.

Digital input

Example

A digital input signal can havethe values0 or 1. The RAPID program can read its value but
cannot set its value.

If the digital input signal di1 is 1 then the robot will move.
IF dil = 1 THEN
MovelL pl0, v1000, fine, tPen;
ENDIF

Digital output

Example

Other signal types

A digital output signal can have the values 0 or 1. The RAPID program can set the value for
adigital output signal, and thus aff ect external equipment. Thevalue of adigital output signal
is set with the instruction setpo.

Therobot hasagrip tool that can be closed with the digital output signal do_grip. Therobot
movesto the position where the pen isand closes the gripper. The robot then movesto where
it shall draw, now using the tool tpen.

Moved p0, vmax, fine, tGripper;

SetDO do grip, 1;

MoveL pl0, v1000, fine, tPen;

Digital signalsare most common and easy to use. If asignal needsto have another value than
Oor 1, there are analog signals and groups of digital signalsthat can have other values. These
types of signals are not covered in this manual .

2 RAPID robot functionality

2.3.1. Communicate with the FlexPendant

2.3 User interaction

2.3.1. Communicate with the FlexPendant

About read and write instructions
In RAPID, there are several instructions for writing information to the robot operator, aswell
as receiving input from the operator. We have already seen Tpwrite in previous examples.
The only instructions we will ook at here are TPwrite, TPReadFK and TPReadNum.

TPWrite
Writing a message to the operator can be made with the instruction Tpwrite.

TPWrite "Now producing exhaust pipes";

I | I\Illli[% ol

All tasks

How producing exhaust pipes

xx0700000374
Write a string variable
Thetext string written on the FlexPendant can come from astring variable, or the written text
can be a concatenation of a string variable and another string.
VAR string product := "exhaust pipe";
! Write only the product on the FlexPendant
TPWrite product;
! Write "Producing" and the product on the FlexPendant

TPWrite "Producing " + product;
Write a numerical variable
A numerical variable can be added after the string using the optional argument \ Num.

VAR num count := 13;

TPWrite "The number of produced units is: " \Num:=count;

Continues on next page

2 RAPID robot functionality

2.3.1. Communicate with the FlexPendant

Continued
TPReadFK
When writing a RAPID program that requires the operator to make a choice, TPrReadFK isa
useful instruction. It allows up to five function keys to be displayed, and the operator can
choose which one to tap. The buttons will correspond to the values 1 to 5.
VAR num answer;
TPReadFK answer, "Select which figure to draw", "Square",
"Triangle", stEmpty, stEmpty, stEmpty;
IF answer = 1 THEN
! code to draw square
ELSEIF answer = 2 THEN
! code to draw triangle
ELSE
! do nothing
ENDIF
|| aADn I[ % ‘[ Manual Motors On E]]
irp | Svstem1(SEVST-w-0003294) Running (Speed 100%)
—
EE-
Select which figure to draw
Square Triangle
e et E3
xx0700000376
If the user selects "Square”, the numeric variable answer getsthe value 1. If the user selects
"Triangle", the numeric variable answer getsthe value 2.
Five functions keys can be specified. If akey isnot being used, write st Empty instead of the
text on the button.
TPReadNum

TPReadNum allows the operator to write a number on the FlexPendant, rather than just
making a choice.
VAR num answer;
TPReadNum answer, "How many times shall I draw the figure?";
FOR i FROM 1 TO answer DO
! code to draw figure
ENDFOR

Continues on next page

2 RAPID robot functionality

2.3.1. Communicate with the FlexPendant

Continued
Ak 1D I. Manual Motors On m
FLipep % ‘ System1({SEVST-W-0003294) Running {Speed 100%)
| (A TFTRelggnglm
How many times shall I draw the figure?
7 8 9 “—
4 5 5] —>
12| 3 |
o | +/
oK
[. T_ROBL : | @ I
! MainModule
xx0700000378
The numeric variable answer gets the value that the operator types.

3 Structure

3 Structure

3.1. RAPID procedure

3.1. RAPID procedure

What is a procedure

So far, al the RAPID code examples we have looked at have only executed code in the
procedure main. The execution automatically startsin the procedure named main, but there
can be several procedures.

A procedure must be declared with the keyword proc followed by the procedure name, the
procedure arguments and the program code that the procedure should execute. A procedure
is called from another procedure (except main, which is automatically called when the
program starts).

Example

If we want to draw four squares of different sizes, we could write amost the same program
code four times. Thiswould result in alot of code and make the program difficult to
understand. A much more efficient way to write this program is to make a procedure that
draws the square, and let the main procedure call this procedure four times.

PERS tooldata tPen:=
CONST robtarget plO:=

PROC main ()
! Call the procedure draw_square
draw_square 100;
draw_square 200;
draw_square 300;
draw_square 400;
ENDPROC

PROC draw_square (num side_size)
VAR robtarget p20;
VAR robtarget p30;
VAR robtarget p40;

! p20 is set to pl0 with an offset on the y value

p20 := Offs(pl0, 0, side size, 0);
p30 := Offs(pl0, side size, side _size, 0);
p40 := Offs(pl0, side size, 0, 0);

MovelL pl0, v200, fine, tPen;

MovelL p20, v200, fine, tPen;

MovelL p30, v200, fine, tPen;

MovelL p40, v200, fine, tPen;

MoveL pl0, v200, fine, tPen;
ENDPROC

Continues on next page

3 Structure

3.1. RAPID procedure

Continued

Procedure arguments

When declaring aprocedure, all argumentsare declared inside parenthesis after the procedure
name. This declaration contains data type and argument name for each argument. The
argument getsitsvalue from the procedure call and the argument acts asavariableinside the
procedure (the argument cannot be used outside its procedure).

PROC main()
my procedure 14, "Hello", TRUE;
ENDPROC

PROC my procedure (num nbr times, string text, bool flag)

ENDPROC

Inside the procedure my procedure above, nbr times hasthevalue 14, text hasthe
value "Hello" and £1ag has the value TRUE.

When calling the procedure, the order of the argumentsisimportant to give theright valueto
the right argument. No parenthesis are used in the procedure call.

Variables declared inside the procedure

Variables declared inside a procedure only exist inside that procedure. |.e. in the example
above, p10 can be used in al proceduresin this module, but p20, p30 and p40o can only be
used in the procedure draw_square.

36

3 Structure

3.2. Modules

About modules

3.2. Modules

A RAPID program can consist of one or several modules. Each module can contain one or
several procedures.

The small and simple programsthat are shown in thismanual only use one module. Inamore
complex programming environment, some standard procedures, used by many different
programs, can be placed in a separate module.

Example

The module MainModule contains code that is specific for this program and specifies what
the robot should do in this particular program. The module figures_module contains
standard code that can be used by any program that wantsto draw a square, triangle or circle.

MODULE MainModule
draw_square;
ENDMODULE

MODULE figures_module
PROC draw_square ()

ENDPROC
PROC draw_triangle()

ENDPROC
PROC draw_circle()

ENDPROC
ENDMODULE

Program modules

System modules

A program module is saved with the file ending .mod, e.g. figures_module.mod.

It makes no difference for the robot controller if the program is written in several modules.
The reason to use several program modulesis only to make the program easier to grasp and
easier to reuse for the programmers.

There can only be one program active on the robot controller, i.e. only one of the modulescan
contain a procedure named main.

A system module is saved with the file ending .sys, e.g. system data_module.sys.

Dataand procedures that should be kept in the system even if the program is changed should
be placed in a system module. For example, if a persistent variable of type tooldata is
declared in asystem module, arecalibration of thetool ispreserved even if anew programis
loaded.

3.3. Structured design

3.3. Structured design

About structure

When first confronting a problem that you want to solve with a RAPID program, sit down
and analyze the problem and its components. If you start programming without first thinking
through the design, your program will beirrational. A well designed program islesslikely to
contain errors and is easier for others to understand. The time spent on design is paid back
many times in testing and maintenance of the program.

Break down the problem

Follow these steps to break down the problem into manageabl e parts:

Action

1. Identify larger functionality. Try to split the problem into smaller pieces that will be easier
to handle.

2. Create a design structure. Draw a map of the functionality and how they relate to each
other.

3. Look at each block in the design structure. Can a block be further split into smaller
pieces? What is required to implement the block?

Example

Problem description

Design structure

Create a RAPID program that can draw squares or triangles on a piece of paper. Let the
operator decideif it isa square or triangle that should be drawn next. When the robot is
finished drawing the figure the user shall be able to make the same selection again until the
operator taps on a Quit button.

When the robot has drawn 10 figures on the same paper, write amessage that the paper should
be replaced and wait for the operator to tap an OK button.

Between drawings, check if di1 is 1. If itis, moveto apencil sharpener and set do1 to 1 to
start the sharpener and slowly move the pencil into the sharpener. Normally we would need
to redefine the tool sinceit gets shorter when it is sharpened, but we will skip that step in this
example.

MAIN
count>=10 di=1
User selection Change paper Pencil sharpener

Square | |Triangle Quit

en0700000381

Continues on next page

3 Structure

3.3. Structured design
Continued

Program code
MODULE MainModule

PERS tooldata tPen := [ TRUE, [[200, O, 30], [1, O, O ,011, [0.8,
(62, o, 171, [1, 0, O, O], O, O, 011;

CONST robtarget pl0 := [ [600, -100, 800], [0.707170, O,
0.707170, 0], [0, O, O, 0], [ 9E9, 9E9, 9E9, 9E9, 9E9, 9E9,
9E9] 1;

CONST robtarget pSharpl := [ [200, 500, 850], [1, O, O, 0], [O,
0, 0, 0], [ 9E9, 9E9, 9E9, 9E9, 9E9, 9E9, 9E9] 1;

PERS num count := 0;

PROC main ()
user selection;
IF count >= 10 THEN
change paper;
! Reset count
count := 0;
ENDIF
IF di=1 THEN
sharpen pencil;
ENDIF
ENDPROC

PROC user_selection()
VAR num answer;
TPReadFK answer, "Select which figure to draw", "Square",
"Triangle", "Quit", stEmpty, stEmpty;
IF answer = 1 THEN
draw_square;
count := count + 1;
ELSEIF answer = 2 THEN
draw_triangle;
count := count + 1;
ELSE
quit;
ENDIF
ENDPROC

Continues on next page

3.3. Structured design

Continued

PROC draw_square ()
VAR robtarget p20;
VAR robtarget p30;
VAR robtarget p40;

! Define points that give a square with the side 200 mm
p20 := Offs(pl0, 0, 200, 0);

P30 Offs (pl0, 200, 200, 0);

P40 Offs (p10, 200, 0, 0);

MovelL pl0, v200, fine, tPen;

MoveL p20, v200, fine, tPen;

MoveL p30, v200, fine, tPen;

MovelL p40, v200, fine, tPen;

MoveL pl0, v200, fine, tPen;
ENDPROC

PROC draw_triangle()
VAR robtarget p20;
VAR robtarget p30;

! Define points for the triangle
p20 := Offs(pl0, 0, 200, 0);
p30 := Offs(pl0, 200, 100, 0);

MovelL pl0, v200, fine, tPen;

MovelL p20, v200, fine, tPen;

MovelL p30, v200, fine, tPen;

MovelL pl0, v200, fine, tPen;
ENDPROC

PROC quit ()
TPWrite "Good bye!™"
! Terminate the program
EXIT;

ENDPROC

PROC change paper ()
VAR num answer;

TPReadFK answer, "Change the paper", "OK", stEmpty, stEmpty,

stEmpty, stEmpty;
ENDPROC

Continues on next page

3 Structure

PROC sharpen pencil ()
VAR robtarget pSharp2;
VAR robtarget pSharp3;

pSharp2 := Offs(pSharpl, 100, 0, 0);
pSharp3 := Offs(pSharpl, 120, 0, 0);

3.3. Structured design

Continued

! Move quickly to position in front of sharpener

Moved pSharpl, vmax, zl1l0, tPen;

! Place pencil in sharpener

MovelL pSharp2, v500, fine, tPen;

! Start the sharpener

SetDO dol, 1;

! Slowly move into the sharpener

Movel pSharp3, v5, fine, tPen;

! Turn off sharpener

SetDO dol, 0;

! Move out of sharpener

MoveL pSharpl, v500, fine, tPen;
ENDPROC

ENDMODULE
Note that in production a program is normally run in continuous mode, so that when the
execution reaches the end of the main procedure it starts from the beginning again. If thisis
not used, awHILE loop can be used to repeat everything inside themain procedure.

3HAC029364-001 Revision: -

3.3. Structured design

4 Data with multiple values

4.1. Arrays

4 Data with multiple values

4.1. Arrays

What is an array

An array is avariable that contains more than one value. An index is used to indicate one of
the values.

Declaring an array

The declaration of an array looks like any other variable, except that the length of thearray is
specified inside{ }.

VAR num my array{3};

Assigning values

Anarray can beassigned al itsvalues at once. When assigning the whole array the values are
surrounded by [ ] and separated by commas.

my array := [5, 10, 7];
Itisalso possibleto assign avalueto one of the elementsin an array. Which element to assign
avalueto is specified inside{ }.

my array{3} := 8;

Example

Thisexample use a For loop and arraysto ask the operator for the estimated production time
for each part. Itisavery efficient way to write code compared to having onevariablefor each
part and not be able to use the For loop.

VAR num time{3};

VAR string part{3} := ["Shaft", "Pipe", "Cylinder"];

VAR num answer;

PROC main ()
FOR i FROM 1 TO 3 DO
TPReadNum answer, "Estimated time for " + part{i} + "?";
time{i} := answer;
ENDFOR
ENDPROC

4.2. Composite data types

4.2. Composite data types

What is a composite data type

A composite datatype is a data type that contains more than one value. It is declared asa
normal variable but contains a predefined number of values.

pos
A simple example of a composite data type isthe datatype pos. It contains three numerical
values (x, y and 2).
The declaration looks like a simple variable:
VAR pos posl;
Assigning al valuesis done like with an array:
posl := [600, 100, 800];
Thedifferent components have namesinstead of numbers. The componentsin pos are named
%,y and z. The value in one component isidentified with the variable name, a point and the
component name:
posl.z := 850;
orient
Thedatatypeorient specifiesthe orientation of thetool. The orientation isspecified by four
numerical values, named g1, g2, g3 and q4.
VAR orient orientl := [1, 0, 0, 0];
TPWrite "The value of gl is " \Num:=orientl.qgl;
pose
A datatype can be composed of other composite data types. An example of thisisthe data
type pose, which consists of one pos named trans and one orient named rot.
VAR pose posel := [[600, 100, 800], [1, 0, 0, 0]];
VAR pos posl := [650, 100, 850];
VAR orient orientl;
posel.pos := posl;
orientl := posel.rot;
posel.pos.z := 875;
robtarget
robtarget iSt00 complex adatatype to explain in detail here, so we will settle for a brief
explanation.
robtarget consists of four parts:
Data type Name Description
pos trans X, y and z coordinates
orient rot Orientation
confdata robconf Specifies robot axes angles
extjoint extax Specifies positions for up to 6 additional axes. The value
is set to 9E9 where no additional axis is used.
Continues on next page

4 Data with multiple values

4.2. Composite data types
Continued

VAR robtarget pl0 := [ [600, -100, 800], [0.707170, 0, 0.707170,
o], f[o, 0, O, 0], [ 9E9, 9E9, 9E9, 9E9, 9E9, 9E9] 1];

! Increase the x coordinate with 50

plO0.trans.x := plO.trans.x + 50;

Detailed descriptions
Detailed descriptions of these data types and many more can be found in Technical reference
manual - RAPID Instructions, Functions and Data types, section Data types.

5 RAPID instructions and functions

5.1. Instructions

5 RAPID instructions and functions

5.1. Instructions

What is an instruction

A RAPID instruction acts as a premade procedure. Aninstruction call looks like a procedure
call with the instruction name followed by argument values.

Some RAPID instructions are simple and could easily have been written as a procedure in
RAPID. For example the instruction add.

Add regl, 3;

! The same functionality could be written:

regl := regl + 3;
Other RAPID instructions perform complicated processes that could not have been
programmed without these premade instructions. For example MoveL, which may seem like
asimpleinstruction but in the background there are cal culations of how much to move each
robot axisand how much current each motor should have. Because the program codefor these
calculationsis already made, all you have to do is write asimple instruction call.

MoveL pl0, v1000, fine, tool0;

Detailed descriptions

Detailed descriptions of instructions can be found in Technical reference manual - RAPID
Instructions, Functions and Data types, section Instructions.

5.2. Functions

5.2. Functions

What is a function
A RAPID function is similar to an instruction but returns avalue.
! Calculate the cosine of reg2
regl : = Cos(reg2);
Since the function returns a value, the result of the function can be assigned to avariable.

The argumentsin afunction call are written inside parenthesis and are separated with
commas.

Include a function call in a statement
Anywhere, where avalue can be used, afunction returning a value of the same datatype can
be used.
! Perform something if regl is smaller than -2 or greater than 2
IF Abs(regl) > 2 THEN

! Convert the num time to string and concatenate with other strings

stringl := name + "’'s time was " + NumToStr (time) ;

Simplify complicated calculations
A single function call can often replace several complex statements.
For example:
p20 := Offs(pl0, 100, 200, 300);
can replace the following code:

p20 := pl0;

p20.trans.x := p20.trans.x + 100;
p20.trans.y := p20.trans.y + 200;
p20.trans.z := p20.trans.z + 300;

Detailed descriptions
Detailed descriptions of functions can be found in Technical reference manual - RAPID
Instructions, Functions and Data types, section Functions.

6 What to read next

6 What to read next

6.1. Where to find more information

6.1. Where to find more information

What to find in which manual

What do you want to know

e How to write programs on the FlexPendant
* How to load programs to the robot controller

+ How to test the program

e More detailed information about the functionality

mentioned in this manual

* Whatinstructions are there for a specific category
(e.g. move instructions or clock functionality)

» Descriptions of more advanced functionality (e.g.

interrupts or error handling)

« Information about a specific instruction, function

or data type

» Details about how the robot controller handles

different parts of RAPID

Where to read about it

Operating manual - IRC5 with
FlexPendant, section
Programming and testing

Technical reference manual -
RAPID overview

Technical reference manual -
RAPID Instructions, Functions and
Data types

Technical reference manual -
RAPID kernel

